Numerical routines for arXiv:1305.1567

“Almost one bit violation for the additivity of the minimum output entropy”, by Serban T. Belinschi, Benoit Collins, Ion Nechita

---

1. Main routines

In[1]:=
```
ClearAll["Global`*"]
```

In[2]:=
```
Needs["PlotLegends`"]
```

In[3]:=
```
phi[x_, y_] := Module[{},
   If[x + y < 1, x + y - 2 * x * y + 2 * Sqrt[(1 - x) * (1 - y) * x * y], 1]
  ];
dif1[k_, t_] := Module[{γ1, γ2, x1, x2, h1, h2},
   γ1 = t + (1 - t) / k^2;
   γ2 = (1 - t) / k^2;
   h1 = -γ1 * Log[γ1] - (k^2 - 1) * γ2 * Log[γ2];
   x1 = phi[1 / k, t];
   x2 = (1 - x1) / (k - 1);
   h2 = -x1 * Log[x1] - (k - 1) * x2 * Log[x2];
   h1 - 2 * h2
  ];
dif1HW[k_, t_] := Module[{γ1, γ2, x1, x2, h1, h2},
   γ1 = t;
   γ2 = (1 - t) / (k^2 - 1);
   h1 = -γ1 * Log[γ1] - (k^2 - 1) * γ2 * Log[γ2];
   x1 = phi[1 / k, t];
   x2 = (1 - x1) / (k - 1);
   h2 = -x1 * Log[x1] - (k - 1) * x2 * Log[x2];
   h1 - 2 * h2
  ];
difp[k_, t_, p_] := Module[{γ1, γ2, x1, x2, h1, h2},
   γ1 = t + (1 - t) / k^2;
   γ2 = (1 - t) / k^2;
   h1 = Log[γ1^p + (k^2 - 1) * γ2^p] / (1 - p);
   x1 = phi[1 / k, t];
   x2 = (1 - x1) / (k - 1);
   h2 = Log[x1^p + (k - 1) * x2^p] / (1 - p);
   h1 - 2 * h2
  ];
difinfty[k_, t_] := Module[{γ1, γ2, x1, x2, h1, h2},
   γ1 = t + (1 - t) / k^2;
   h1 = -Log[γ1];
   x1 = phi[1 / k, t];
   h2 = -Log[x1];
   h1 - 2 * h2
  ];
```

2. Trigonometric substitution in the derivative of eq. (40)

In[8]:=
```
func = (phi[x, t] / x - (1 - phi[x, t]) / (1 - x)) / D[phi[x, t], x] -
  Log[phi[x, t] / x] + Log[(1 - phi[x, t]) / (1 - x)]
```

Out[8]=
$$\left(-1/(1-x)\left(1-\text{If}\left[t+x<1,\ x+t-2xt+2\sqrt{(1-x)(1-t)xt},\ 1\right]\right)+1/x\,\text{If}\left[t+x<1,\ x+t-2xt+2\sqrt{(1-x)(1-t)xt},\ 1\right]\right)\Big/ \text{If}\left[t+x<1,\ 1-2t+\frac{(1-x)(1-t)t-(1-t)xt}{\sqrt{(1-t)t(1-x)x}},\ 0\right]+\text{Log}\left[\frac{1}{1-x}\left(1-\text{If}\left[t+x<1,\ x+t-2xt+2\sqrt{(1-x)(1-t)xt},\ 1\right]\right)\right]-\text{Log}\left[\frac{1}{x}\text{If}\left[t+x<1,\ x+t-2xt+2\sqrt{(1-x)(1-t)xt},\ 1\right]\right]$$

In[9]:=
```
Simplify[D[func, x] /. {t → Cos[r]^2, x → Sin[s]^2}, {0 < r, r < Pi / 2, 0 < s, s < r}]
```

Out[9]=
$$\begin{cases} \text{Indeterminate} & \text{Cos}[r]^2+\text{Sin}[s]^2\geq 1 \\ -16\,\text{Cos}[r]^3\,\text{Csc}[2r-2s]^2\,\text{Csc}[2s]^3\,\text{Sin}[r-2s] & \text{True} \end{cases}$$

3. MOE violations

In[10]:=
```
Plot[{dif1[181, t], dif1[182, t], dif1[183, t], dif1[184, t]}, {t, 0, 1},
 PlotLegend → {"k=181", "k=182", "k=183", "k=184"}, LegendPosition → {1.1, -0.4}]
```

Out[10]=



In[11]:=
```
Plot[{dif1[180, t], dif1[181, t], dif1[182, t],
  dif1[183, t], dif1[184, t], dif1[185, t]}, {t, 0.1, 0.2},
 PlotLegend → {"k=180", "k=181", "k=182", "k=183", "k=184", "k=185"},
 LegendPosition → {1.1, -0.4}]
```

Out[11]=



In[12]:=
```
NMinimize[{dif1[182, t], 0 < t < 1}, t]
NMinimize[{dif1[183, t], 0 < t < 1}, t]
```

Out[12]=
```
{0.000435891, {t → 0.110163}}
```

Out[13]=
```
{-0.000122982, {t → 0.112406}}
```

4. MOE violations with Hayden-Winter bound

In[14]:=
```
Plot[{dif1HW[180, t], dif1HW[181, t], dif1HW[182, t],
  dif1HW[183, t], dif1HW[184, t], dif1HW[185, t]}, {t, 0, 1},
 PlotLegend → {"k=180", "k=181", "k=182", "k=183", "k=184", "k=185"},
 LegendPosition → {1.1, -0.4}]
```

Out[14]=



In[15]:=
```
Plot[{dif1HW[180, t], dif1HW[181, t], dif1HW[182, t],
  dif1HW[183, t], dif1HW[184, t], dif1HW[185, t]}, {t, 0.1, 0.2},
 PlotLegend → {"k=180", "k=181", "k=182", "k=183", "k=184", "k=185"},
 LegendPosition → {1.1, -0.4}]
```

Out[15]=



In[16]:=
```
Plot[{dif1HW[180, t], dif1HW[181, t], dif1HW[182, t],
  dif1HW[183, t], dif1HW[184, t], dif1HW[185, t]}, {t, 0.1, 0.14},
 PlotLegend → {"k=180", "k=181", "k=182", "k=183", "k=184", "k=185"},
 LegendPosition → {1.1, -0.4}]
```

Out[16]=



In[17]:=
```
NMinimize[{dif1HW[182, t], 0 < t < 1}, t]
NMinimize[{dif1HW[183, t], 0 < t < 1}, t]
NMinimize[{dif1HW[184, t], 0 < t < 1}, t]
```

Out[17]= {0.000659372, {t → 0.110149}}

Out[18]= {0.0000983977, {t → 0.112396}}

Out[19]= {-0.000467132, {t → 0.114597}}

5. MOE violations for t=1/2

In[20]:=
```
Plot[dif1[k, 1 / 2], {k, 2, 500}]
```

Out[20]=



In[21]:=
```
FindRoot[dif1[k, 1 / 2], {k, 100}]
```

Out[21]=
```
{k → 275.801}
```

6. MOE violations for t=1/k

In[22]:=
```
Plot[dif1[k, 1 / k], {k, 2, 1000}]
```

Out[22]=



In[23]:=
```
FindRoot[dif1[k, 1 / k], {k, 100}]
```

Out[23]=
```
{k → 431.819}
```

7. MOE violations at p=2

In[24]:=
```
Plot[{difp[14, t, 2], difp[15, t, 2], difp[16, t, 2], difp[17, t, 2]}, {t, 0, 1},
 PlotLegend → {"k=14", "k=15", "k=16", "k=17"}, LegendPosition → {1.1, -0.4}]
```

Out[24]=



8. MOE violations at p=3

In[25]:=
```
Plot[{difp[12, t, 3], difp[13, t, 3], difp[14, t, 3], difp[15, t, 3]}, {t, 0, 1},
 PlotLegend → {"k=12", "k=13", "k=14", "k=15"}, LegendPosition → {1.1, -0.4}]
```

Out[25]=



9. MOE violations at p=4

In[26]:=
```
Plot[{difp[11, t, 4], difp[12, t, 4], difp[13, t, 4], difp[14, t, 4]}, {t, 0, 1},
 PlotLegend → {"k=11", "k=12", "k=13", "k=14"}, LegendPosition → {1.1, -0.4}]
```

Out[26]=



10. MOE violations at p=∞

In[27]:=
```
Plot[{difinfty[11, t], difinfty[12, t], difinfty[13, t], difinfty[14, t]}, {t, 0, 1},
 PlotLegend → {"k=11", "k=12", "k=13", "k=14"}, LegendPosition → {1.1, -0.4}]
```

Out[27]=